


Lecture 8: Intro to Assembly Programming

The MIPS Microprocessor



Intro to Machine Code

- Now that we have a processor, operations are performed by:
 - The instruction register sends instruction components to the control unit.
 - The control unit decodes instruction according to the **opcode** in the first 6 bits.
 - The control unit sending a sequence of signals to the rest of the processor.
- Only questions remaining:
 - Where do these instructions come from?
 - How are they provided to the instruction memory?


A little about MIPS

- MIPS

- Short for **M**icroprocessor without **I**nterlocked **P**ipeline **S**tages
 - A type of **RISC** (**R**educed **I**nstruction **S**et **C**omputer) architecture.
- Provides a set of simple and fast instructions
 - Compiler translates instructions into 32-bit instructions for instruction memory.
 - Complex instructions are built out of simple ones by the compiler and assembler.


MIPS Memory and Instructions

- All memory is addressed in bytes.
- Instruction addresses are measured in bytes, starting from the instruction at address 0.
- All instructions are 32 bits (4 bytes) long
- Therefore:
all instruction addresses are divisible by 4.




Recall: MIPS instruction types

- **R-type:**



- **I-type:**



- **J-type:**




MIPS Registers

- In MIPS is **register-to-register** (a.k.a. **load-store**) architecture
 - Source, destination of ALU operations are registers.
- MIPS provides 32 registers.
 - Some have special values:
 - Register 0 (**\$zero**): value 0 – always (writes to it are discarded)
 - Register 1 (**\$at**): reserved for the assembler.
 - Registers 28–31 (**\$gp, \$sp, \$fp, \$ra**): memory and function support
 - Registers 26–27: reserved for OS kernel
 - Some are used by programs as functions parameters:
 - Registers 2–3 (**\$v0, \$v1**): return values
 - Registers 4–7 (**\$a0-\$a3**): function arguments
 - Some are used by programs to store values:
 - Registers 8–15, 24–25 (**\$t0-\$t9**): temporaries
 - Registers 16–23 (**\$s0-\$s7**): saved temporaries
 - Also three special registers (**PC, HI, LO**) that are not directly accessible.
 - HI and LO are used in multiplication and division, and have special instructions for accessing them.

Assembly Language Introduction

```
loop: lw    $t3, 0($t0)
      lw    $t4, 4($t0)
      add   $t2, $t3, $t4
      sw    $t2, 8($t0)
      addi  $t0, $t0, 4
      addi  $t1, $t1, -1
      bgtz  $t1, loop
```



Assembler

```
0x8d0b0000
0x8d0c0004
0x016c5020
0xad0a0008
0x21080004
0x2129ffff
0x1d20fff9
```

Assembly vs Machine Code

- Each processor type has its own language for representing 32-bit instructions as user-readable code words.

- Example: $C = A + B$

- Assume A is stored in $\$t1$, B in $\$t2$, C in $\$t3$.

- **Assembly language** instruction:

```
add $t3, $t1, $t2
```

- **Machine code** instruction:


```
000000 01001 01010 01011 XXXXX 100000
```

Note: There is a 1-to-1 mapping for all assembly code and machine code instructions!



Assembly language

- Assembly language is the lowest-level language that you'll ever program in.
- Many compilers translate their high-level program commands into assembly commands, which are then converted into machine code and used by the processor.
- Note: There are multiple types of assembly language, especially for different architectures!



Why learn assembly?


- Understand how code really works
- Better analyze code (runtime, control flows, pointers, stack overflows)
- Make you appreciate constructs of high level languages
- Connect your high level programming knowledge to hardware
- It's on the exam...

Arithmetic instructions

Instruction	Opcode/Function	Syntax	Operation
add	100000	\$d, \$s, \$t	\$d = \$s + \$t
addu	100001	\$d, \$s, \$t	\$d = \$s + \$t
addi	001000	\$t, \$s, i	\$t = \$s + SE(i)
addiu	001001	\$t, \$s, i	\$t = \$s + SE(i)
div	011010	\$s, \$t	lo = \$s / \$t; hi = \$s % \$t
divu	011011	\$s, \$t	lo = \$s / \$t; hi = \$s % \$t
mult	011000	\$s, \$t	hi:lo = \$s * \$t
multu	011001	\$s, \$t	hi:lo = \$s * \$t
sub	100010	\$d, \$s, \$t	\$d = \$s - \$t
subu	100011	\$d, \$s, \$t	\$d = \$s - \$t


Note: "hi" and "lo" refer to the high and low bits referred to in the register slide.
"SE" = "sign extend".

Assembly → Machine Code



Instruction	Opcode/Function	Syntax	Operation
add	100000	\$d, \$s, \$t	\$d = \$s + \$t

R-type instruction!




Although we specify “don’t care” bits as X values, the assembler generally assigns some value (like 0).

Logical instructions

Instruction	Opcode/Function	Syntax	Operation
and	100100	\$d, \$s, \$t	\$d = \$s & \$t
andi	001100	\$t, \$s, i	\$t = \$s & ZE(i)
nor	100111	\$d, \$s, \$t	\$d = ~(\$s \$t)
or	100101	\$d, \$s, \$t	\$d = \$s \$t
ori	001101	\$t, \$s, i	\$t = \$s ZE(i)
xor	100110	\$d, \$s, \$t	\$d = \$s ^ \$t
xori	001110	\$t, \$s, i	\$t = \$s ^ ZE(i)

Note: ZE = zero extend (pad upper bits with 0 value).

Assembly → Machine Code II



Instruction	Opcode/Function	Syntax	Operation
andi	001100	\$t, \$s, i	\$t = \$s & ZE(i)

I-type instruction!



Shift instructions

Instruction	Opcode/Function	Syntax	Operation
sll	000000	\$d, \$t, a	\$d = \$t << a
sllv	000100	\$d, \$t, \$s	\$d = \$t << \$s
sra	000011	\$d, \$t, a	\$d = \$t >> a
srav	000111	\$d, \$t, \$s	\$d = \$t >> \$s
srl	000010	\$d, \$t, a	\$d = \$t >>> a
srlv	000110	\$d, \$t, \$s	\$d = \$t >>> \$s

Note: srl = "shift right logical"

sra = "shift right arithmetic".

The "v" denotes a variable number of bits, specified by \$s. a is **shift amount**, and is stored in **shamt** when encoding the R-type machine code instructions.

Data movement instructions

Instruction	Opcode/Function	Syntax	Operation
<code>mfhi</code>	010000	<code>\$d</code>	<code>\$d = hi</code>
<code>mflo</code>	010010	<code>\$d</code>	<code>\$d = lo</code>
<code>mthi</code>	010001	<code>\$s</code>	<code>hi = \$s</code>
<code>mtlo</code>	010011	<code>\$s</code>	<code>lo = \$s</code>

- These are instructions for operating on the HI and LO registers described earlier (for multiplication and division)

ALU instructions in RISC

- Most ALU instructions are R-type instructions.
 - The six-digit codes in the tables are therefore the function codes (opcodes are 000000).
 - Exceptions are the I-type instructions (`addi`, `andi`, `ori`, etc.)
- Not all R-type instructions have an I-type equivalent.
 - RISC principle dictate that **an operation doesn't need an instruction if it can be performed through multiple existing operations.**
 - Example: `addi + div → divi`

Pseudoinstructions

- Move data from `$t4` to `$t5`?
 - `move $t5,$t4` →

```
add $t5,$t4,$zero
```

- Multiply and store in `$s1`?
 - `mul $s1,$t4,$t5` →

```
mult $t4,$t5  
mflo $s1
```

Time for a Break



Making an assembly program

- Assembly language programs typically have structure similar to **simple** Python or C programs:
 - They set aside registers to store data.
 - They have sections of instructions that manipulate this data.
- It is always good to decide at the beginning which registers will be used for what purpose!
 - More on this later 😊

Time to write our
first assembly program



Compute $\text{result} = a^2 + 2b + 10$

- Set up values in registers
 - $a \rightarrow \$t0, b \rightarrow \$t1$
 - $\text{temp} \rightarrow \$t6$
- $\text{temp} = 10$
- $\text{temp} = \text{temp} + b$
- $\text{temp} = \text{temp} + b$ (again!)
- $\text{result} = a * a$
- $\text{result} = \text{result} + \text{temp}$

```
addi $t0, $zero, 7
addi $t1, $zero, 9

addi $t6, $zero, 10

add $t6, $t6, $t1
add $t6, $t6, $t1

mult $t0, $t0
mflo $t4
mfhi $t5

add $t4, $t4, $t6
```


Formatting Assembly Code

- Instructions are written
`as: <instr> <parameters>`
- Each instruction is written on its own line
- 3 columns
 - Labels
 - Code
 - Comments
- Start with `.text` (we'll see other options later)
- First line of code to run = `label: main`

```
# Compute the following result:  $r = a^2 + 2b + 10$ 
```

```
.text
```

```
# load up some values to test
```

```
main: addi $t0, $zero, 7
```

```
      addi $t1, $zero, 9
```

```
# $t0 will be a,    $t1 will be b,    $t5:$t4 will be r
```

```
# $t6 will be temp
```

```
      addi $t6, $zero, 10 # add 10 to r
```

```
      add $t6, $t6, $t1   # then add b
```

```
      add $t6, $t6, $t1   # then add b again
```

```
      mult $t0, $t0       # multiply a * a
```

```
      mflo $t4           # move the low result of  $a^2$ 
```

```
                        # into the low register of r
```

```
      mfhi $t5           # move the high result of  $a^2$ 
```

```
                        # into the high register of r
```

```
      add $t4, $t4, $t6   # add the temporary value
```

```
                        # ( $2b + 10$ ) to the low
```

```
                        # register of r
```

Simulating MIPS

aka: QtSpim

The screenshot displays the QtSpim MIPS simulator interface. The window title is "QtSpim". The menu bar includes "File", "Simulator", "Registers", "Text Segment", "Data Segment", "Window", and "Help".

The "Registers" tab is active, showing the following values:

```
FP Regs [16]
PC = 400048
SPC = 0
Cause = 0
Bad/Addr = 0
Status = 3000ff10
HI = 0
LO = 0
RD [rd] = 0
R1 [at] = 10010000
R2 [v0] = a
R3 [v1] = 0
R4 [a0] = 3
R5 [a1] = 7ffff694
R6 [a2] = 7ffff6a4
R7 [a3] = 0
R8 [t0] = 2345
R9 [t1] = 33667
R10 [t2] = 2245
R11 [t3] = 33767
R12 [t4] = 0
R13 [t5] = 0
R14 [t6] = 0
R15 [t7] = 0
R16 [a0] = 0
R17 [a1] = 0
R18 [a2] = 0
R19 [a3] = 0
R20 [a4] = 0
R21 [a5] = 0
R22 [a6] = 0
R23 [a7] = 0
R24 [t8] = 0
R25 [t9] = 0
R26 [k0] = 0
R27 [k1] = 0
```

The "Data" tab is active, showing memory contents:

```
User data segment [10000000]..[10040000]
[10000000]..[1000ffff] 00000000
[10010000] 00002345 00033667 00033767 00000000 E# . . g 6 . . g 7 . . . . .
[10010010]..[1003ffff] 00000000 I

User Stack [7ffff690]..[80000000]
[7ffff690] 00000003 7ffff79b 7ffff790 7ffff779 . . . . . Y . . .
[7ffff6a0] 00000000 7ffff7d6 7ffff7b7 7ffff791 . . . . . . . . .
[7ffff6b0] 7ffff65a 7ffff71e 7ffff7ed 7ffff7eb 2 . . . . . . . . .
[7ffff6c0] 7ffff69c 7ffff778 7ffff7e1 7ffff7e3d . . . . . x . . . Q . . . = . . .
[7ffff6d0] 7ffff630 7ffff71b 7ffff7e0 7ffff7e08 0 . . . . . . . . .
[7ffff6e0] 7ffff60e 7ffff7d5 7ffff7e3 7ffff7d2 . . . . . . . . .
[7ffff6f0] 7ffff60b 7ffff7d4 7ffff7e2 7ffff7d7 . . . . . . . . . W . . .
[7ffff700] 7ffff63c 7ffff7b1 7ffff7ad 7ffff7a5 < . . . . . . . . .
[7ffff710] 7ffff6ad 7ffff692 7ffff7e6 7ffff7e45 . . . . . n . . . E . . .
[7ffff720] 7ffff627 7ffff696 7ffff7e5 7ffff797 . . . . . . . . .
[7ffff730] 7ffff688 7ffff694 7ffff7e1 7ffff7e32 . . . . . . . . .
[7ffff740] 7ffff69c 7ffff691 7ffff7e1 7ffff7e0e < . . . . . . . . .
[7ffff750] 7ffff69c 7ffff697 7ffff680 7ffff684 . . . . . g . . . P . . . B . . .
[7ffff760] 7ffff628 7ffff616 7ffff7e6 7ffff7b8 (. . . . . . . . .
[7ffff770] 00000000 00000000 83303300 6562136 . . . . . . . . . 3 0 5 6 / m e
[7ffff780] 7972666d 646178e5 2e656c70 00627361 m o r y e s a m p l e . a a m
[7ffff790] 676e6353 45276c6f 43004543 7355273a S c h o o l / E C E . C i : / U s
[7ffff7a0] 2e737265 6e686e6a 636f442f 6e656d75 e r s / J o h n / D o c u m e n
[7ffff7b0] 4a2f7374 006e886f 646e6977 5c73776e t e / J o h n . w i n d o w s _
[7ffff7c0] 63617274 5c7e6e69 6e676d6e 3a656e69 t r a c i n g _ l o g . f i l e =
[7ffff7d0] 425c3a43 69425456 65545c6e 5c737473 C i \ B V T B i n \ T e a s a \
[7ffff7e0] 74736e69 706c6c61 616b6361 635c6567 i n s t a l l p a c k a g e \ c
[7ffff7f0] 6f6c6e73 6c696667 6f6c2e65 69770067 s i l l o g . f i l e . l o g . v i
[7ffff800] 776164ee 72745f73 6e696361 6c665167 n d o w s _ l r a c i n g _ f i
[7ffff810] 3a737671 43770253 7264646e 5c3a3d3d a g e = 3 _ w i n d i r = C \ \
[7ffff820] 646e6957 0073776e 52455355 46425250 W i n d o w s . U S E R R R O F
[7ffff830] 3d454c49 555c3a43 79726573 686f4a5c I L E = C : \ U s e r s \ J o h
[7ffff840] 5355006e 414e5245 4a3d454d 006e886f n . U S E R N A M E = J o h n .
[7ffff850] 52455355 41444f44 4a3d4e49 2d6e886f U S E R D O M A I N = J o h n -
[7ffff860] 7470616c 5400706e 4f435456 4e4f4d4d l a p t o p . T V T C O M M O N
```


The "Text" tab is active, showing assembly code:

```
Memory and registers cleared
Loaded: C:/Users/John/AppData/Local/Temp/qt_temp.116972
SPIM Version 3.1.9 of January 19, 2013
Copyright 1990-2012, James R. Larus.
All Rights Reserved.
SPIM is distributed under a BSD license.
See the file README for a full copyright notice.
```

QtSpim Simulator

- Link to download:
 - <http://spimsimulator.sourceforge.net>
- MIPS settings in the simulator:
 - From menu, Simulator → Settings
 - Important to **not** have “delayed branches” or “delayed loads” selected under Settings.
 - “Load exception handler” field should also be unchecked.
- You should view user code (Text Segment -> User text), no need for “kernel text”

QtSpim Settings



QtSpim – Quick How To

- Write a MIPS program (similar to the ones posted) in any text editor. Save it with `.asm` extension.
- In QtSpim select:
 - File -> **Reinitialize and load a file**
 - Single step through your program while observing (a) the Int Regs window and (b) the text window (user text).
 - As you step through, the highlighted instruction is the one about to be executed.

QtSpim Help => MIPS reference

- QtSpim help (Help -> View Help) contains
 - “Appendix A (Assemblers, Linkers, and the SPIM Simulator)”
from *Patterson and Hennessey, Computer Organization and Design: The Hardware/Software Interface, Third Edition*
 - Useful reference for MIPS R2000 Assembly Language
 - Look at “Arithmetic and Logical Instructions”.


$$r = (2a + 5) * (7b)$$

```
.text
# $t0 = a, $t1 = b, $t4 = r
# $t7 = left side, $t8 = right side
main:  addi $t0, $zero, 7 # load up some values to test
      addi $t1, $zero, 9
# calculate left side
calc_left:  add $t7, $t0, $t0 # ls <- 2a
           addi $t7, $t7, 5 # ls <- ls + 5

# calculate right side
calc_right: addi $t8, $zero, 7 # rs <- 7
           mult $t8, $t1 # multiply b * 7
           mflo $t8 # put result back into rs

# multiply left * right and put result into r
multiply:  mult $t7, $t8
           mflo $t4
```



Control Flow



Control flow in assembly

- Not all programs follow a linear set of instructions.
 - Some operations require the code to branch to one section of code or another (if/else).
 - Some require the code to jump back and repeat a section of code again (for/while).
- For this, we have **labels** on the left-hand side that indicate the points that the program flow might need to jump to.
 - References to these points in the assembly code are resolved by the assembler at compile time to offset values for the program counter.

Time for more instructions!



Branch instructions

Instruction	Opcode/Function	Syntax	Operation
beq	000100	\$s, \$t, label	if (\$s == \$t) pc \leftarrow label
bgtz	000111	\$s, label	if (\$s > 0) pc \leftarrow label
blez	000110	\$s, label	if (\$s <= 0) pc \leftarrow label
bne	000101	\$s, \$t, label	if (\$s != \$t) pc \leftarrow label

- Branch operations are key when implementing if statements and while loops.
- The labels are memory locations, assigned to each label at compile time.

Branch instructions

- How does a branch instruction work?

```
.text
```

```
main:    beq $t0, $t1, end    # check if $t0 == $t1
        ...                # if $t0 != $t1, then
        ...                # execute these lines

end:     ...                # if $t0 == $t1, then
        ...                # execute these lines
```

Branch instructions

- Alternate implementation using `bne`:

```
.text  
  
main:    bne $t0, $t1, end    # check if $t0 == $t1  
        ...                # if $t0 == $t1, then  
        ...                # execute these lines  
  
end:     ...                # if $t0 != $t1, then  
        ...                # execute these lines
```

- Used to produce `if` statement behaviour.

Conditional Branch Terms

- When the branch condition is met, we say the **branch is taken**.
- When the branch condition is not met, we say the **branch is not taken**.
 - What is the next PC in this case?
 - It's the usual $PC+4$
- How far can a processor branch? Are there any constraints?

Jump instructions

Instruction	Opcode/Function	Syntax	Operation
<code>j</code>	000010	label	$pc \leftarrow \text{label}$
<code>jal</code>	000011	label	$\$ra = pc; pc \leftarrow \text{label}$
<code>jalr</code>	001001	$\$s$	$\$ra = pc; pc = \s
<code>jr</code>	001000	$\$s$	$pc = \$s$

- `jal` = “jump and link”.
 - Register $\$31$ (aka $\$ra$) stores the address that’s used when returning from a subroutine.
- Note: `jr` and `jalr` are not j-type instructions.

Comparison instructions

Instruction	Opcode/Function	Syntax	Operation
slt	101010	\$d, \$s, \$t	\$d = (\$s < \$t)
sltu	101001	\$d, \$s, \$t	\$d = (\$s < \$t)
slti	001010	\$t, \$s, i	\$t = (\$s < SE(i))
sltiu	001001	\$t, \$s, i	\$t = (\$s < SE(i))

- "slt" = "Set Less Than"
- Comparison operation stores a one in the destination register if the less-than comparison is true, and stores a zero in that location otherwise.
- Signed: 0x80000000 is less than all numbers
- Unsigned: 0 - 0x7FFFFFFF are less than 0x80000000

If/Else statements in MIPS

```
if ( i == j )
    i++;
else
    j--;
j += i;
```

- Strategy for if/else statements:
 - Test condition, and jump to `if` logic block whenever condition is true.
 - Otherwise, perform `else` logic block, and jump to first line after `if` logic block.

Translated if/else statements

```
# $t1 = i, $t2 = j
main:    beq  $t1, $t2, IF      # branch if ( i == j )
        addi $t2, $t2, -1    # j--
        j  END              # jump over IF
IF:      addi $t1, $t1, 1     # i++
END:     add  $t2, $t2, $t1   # j += i
```

- Alternately, you can branch on the else condition first:


```
# $t1 = i, $t2 = j
main:    bne  $t1, $t2, ELSE   # branch if ! ( i == j )
        addi $t1, $t1, 1     # i++
        j  END              # jump over ELSE
ELSE:    addi $t2, $t2, -1    # j--
END:     add  $t2, $t2, $t1   # j += i
```

A trick with if statements

- Use flow charts to help you sort out the control flow of the code:

```
if ( i == j )  
    i++;  
else  
    j--;  
j += i;
```

```
# $t1 = i, $t2 = j  
main:    beq  $t1, $t2, IF  
        addi $t2, $t2, -1  
        j  END  
IF:     addi $t1, $t1, 1  
END:    add  $t2, $t2, $t1
```



Multiple Conditions Inside If

```
if ( i == j || i == k )  
    i++ ; // if-body  
else  
    j-- ; // else-body  
j = i + k ;
```

Multiple Conditions Inside If

```
if ( i == j || i == k )
    i++ ; // if-body
else
    j-- ; // else-body
j = i + k ;
```

- Branch statement for each condition:

```
# $t1 = i, $t2 = j, $t3 = k
main:  beq $t1, $t2, IF      # cond1: branch if ( i == j )
      bne $t1, $t3, ELSE   # cond2: branch if ( i != k )
IF:    addi $t1, $t1, 1    # if (i==j|i==k) → i++
      j END              # jump over else
ELSE:  addi $t2, $t2, -1   # else-body: j--
END:   add $t2, $t1, $t3   # j = i + k
```

Multiple if conditions

- How would this look if the condition changed?

```
if ( i == j && i == k )
    i++ ; // if-body
else
    j-- ; // else-body
j = i + k ;
```

```
# $t1 = i, $t2 = j, $t3 = k
main:  bne $t1, $t2, ELSE    # cond1: branch if ( i != j )
      bne $t1, $t3, ELSE    # cond2: branch if ( i != k )
IF:    addi $t1, $t1, 1     # if (i==j|i==k) → i++
      j END                # jump over else
ELSE:  addi $t2, $t2, -1    # else-body: j--
END:   add $t2, $t1, $t3    # j = i + k
```

```
main:    add $t0, $zero, $zero
         addi $t1, $zero, 100
START:   beq $t0, $t1, END
         addi $t0, $t0, 1
         j  START
END:
```


Loops in MIPS

- Example of a simple loop, in assembly:

```
main:    add $t0, $zero, $zero
         addi $t1, $zero, 100
START:   beq $t0, $t1, END
         addi $t0, $t0, 1
         j  START
END:
```

- ...which is the same as saying (in C):

```
int i = 0;
while (i < 100) {
    i++;
}
```

Loops in MIPS

```
for ( <init> ; <cond> ; <update> ) {  
    <for body>  
}
```

- For loops (such as above) are usually implemented with the following structure:

```
main:    <init>  
START:  if (!<cond>) branch to END  
        <for-body>  
UPDATE: <update>  
        jump to START  
END:
```

Loop example in MIPS

```
j = 0;
for ( i=0 ; i<100 ; i++ ) {
    j = j + i;
}
```

- This translates to:

```
# $t0 = i, $t1 = j
main:    add $t0, $zero, $zero      # set i to 0
        add $t1, $zero, $zero    # set j to 0
        addi $t9, $zero, 100     # set $t9 to 100
START:  beq $t0, $t9, EXIT        # branch if i==100
        add $t1, $t1, $t0        # j = j + i
UPDATE: addi $t0, $t0, 1         # i++
        j START
EXIT:
```

- `while` loops are the same, without the initialization and update sections.

Homework

- Fibonacci sequence:
 - How would you convert this into assembly?

```
int n = 10;
int f1 = 1, f2 = 1;

while (n != 0) {
    f1 = f1 + f2;
    f2 = f1 - f2;
    n = n - 1;
}
# result is f1
```